

# NAG Fortran Library Routine Document

## F07GJF (SPPTRI/DPPTRI)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07GJF (SPPTRI/DPPTRI) computes the inverse of a real symmetric positive-definite matrix  $A$ , where  $A$  has been factorized by F07GDF (SPPTRF/DPPTRF), using packed storage.

### 2 Specification

```

SUBROUTINE F07GJF(UPLO, N, AP, INFO)
ENTRY      spptri (UPLO, N, AP, INFO)
INTEGER   N, INFO
real    AP(*)
CHARACTER*1 UPLO

```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

To compute the inverse of a real symmetric positive-definite matrix  $A$ , this routine must be preceded by a call to F07GDF (SPPTRF/DPPTRF), which computes the Cholesky factorization of  $A$  using packed storage.

If UPLO = 'U',  $A = U^T U$  and  $A^{-1}$  is computed by first inverting  $U$  and then forming  $(U^{-1})(U^{-1})^T$ .

If UPLO = 'L',  $A = LL^T$  and  $A^{-1}$  is computed by first inverting  $L$  and then forming  $(L^{-1})^T(L^{-1})$ .

### 4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

### 5 Parameters

- 1: UPLO – CHARACTER\*1 *Input*  
*On entry:* indicates whether  $A$  has been factorized as  $U^T U$  or  $LL^T$  as follows:  
     if UPLO = 'U',  $A = U^T U$ , where  $U$  is upper triangular;  
     if UPLO = 'L',  $A = LL^T$ , where  $L$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: AP(\*) – *real* array *Input/Output*  
**Note:** the dimension of the array AP must be at least  $\max(1, N * (N + 1) / 2)$ .

*On entry:* the upper triangular matrix  $U$  stored in packed form if UPLO = 'U' or the lower triangular matrix  $L$  stored in packed form if UPLO = 'L', as returned by F07GDF (SPPTRF/DPPTRF).

*On exit:*  $U$  is overwritten by the upper triangle of  $A^{-1}$  if UPLO = 'U';  $L$  is overwritten by the lower triangle of  $A^{-1}$  if UPLO = 'L'. More precisely, the  $(i, j)$ th element of  $A^{-1}$  is stored in AP( $i + j(j - 1)/2$ ) for  $i \leq j$  if UPLO = 'U', and in AP( $i + (2n - j)(j - 1)/2$ ) for  $i \geq j$  if UPLO = 'L'.

4: INFO – INTEGER

*Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ , the  $i$ th diagonal element of the Cholesky factor is zero; the Cholesky factor is singular and the inverse of  $A$  cannot be computed.

## 7 Accuracy

The computed inverse  $X$  satisfies

$$\|XA - I\|_2 \leq c(n)\epsilon\kappa_2(A) \quad \text{and} \quad \|AX - I\|_2 \leq c(n)\epsilon\kappa_2(A),$$

where  $c(n)$  is a modest function of  $n$ ,  $\epsilon$  is the *machine precision* and  $\kappa_2(A)$  is the condition number of  $A$  defined by

$$\kappa_2(A) = \|A\|_2 \|A^{-1}\|_2.$$

## 8 Further Comments

The total number of floating-point operations is approximately  $\frac{2}{3}n^3$ .

The complex analogue of this routine is F07GWF (CPPTRI/ZPPTRI).

## 9 Example

To compute the inverse of the matrix  $A$ , where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix}.$$

Here  $A$  is symmetric positive-definite, stored in packed form, and must first be factorized by F07GDF (SPPTRF/DPPTRF).

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07GJF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX
PARAMETER       (NMAX=8)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER       UPLO
*      .. Local Arrays ..
real           AP(NMAX*(NMAX+1)/2)
*      .. External Subroutines ..
EXTERNAL        sptrf, spptri, X04CCF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07GJF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
          READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
          READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
      END IF
*
*      Factorize A
*
      CALL sptrf(UPLO,N,AP,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*      Compute inverse of A
*
          CALL spptri(UPLO,N,AP,INFO)
*
*      Print inverse
*
          IFAIL = 0
*
          CALL X04CCF(UPLO,'Nonunit',N,AP,'Inverse',IFAIL)
*
      ELSE
          WRITE (NOUT,*) 'A is not positive-definite'
      END IF
      END IF
      STOP
*
      END

```

## 9.2 Program Data

```

F07GJF Example Program Data
  4                               :Value of N
  'L'                             :Value of UPLO
  4.16
 -3.12   5.03
  0.56  -0.83   0.76
 -0.10   1.18   0.34   1.18   :End of matrix A

```

### 9.3 Program Results

F07GJF Example Program Results

Inverse

	1	2	3	4
1	0.6995			
2	0.7769	1.4239		
3	0.7508	1.8255	4.0688	
4	-0.9340	-1.8841	-2.9342	3.4978

---